

Komputer sebagai Media Pembelajaran Matematika

Penggunaan MAPLE untuk Pembelajaran KALKULUS

Drs. Sahid, MSc.

Lab Komputer Jurdik Matematika FMIPA UNY

Bagian I: Kalkulus Diferensial

A. Tentang Limit

Pada bagian ini akan dijelaskan bagaimana konsep limit secara numerik, grafik, dan simbolik untuk memperoleh pemahaman yang lebih baik tentang konsep dan aplikasi limit. Untuk itu, jalankan program Maple dan tulis perintah sebagai berikut untuk mengawalinya.

> restart; with(plots):

1. Tinjauan ulang Fungsi & Plot

Format pendefinisian sebuah fungsi di dalam Maple semula mungkin tampak membingungkan ditinjau dari bahasa komputer, namun sesungguhnya notasi Maple sangat sesuai dengan konsep matematik. Di dalam Maple komponen-komponen di dalam definisi sebuah fungsi meliputi:

(1) nama fungsi, (2) tanda := (titik dua dan sama dengan), (3) nama-nama variabel bebas, (4) tanda panah dengan simbol strip dan lebih besar (->), (5) rumus fungsi, dan (6) tanda titik koma (;).


Perhatikan contoh di bawah ini, untuk mendefinisikan fungsi $f(x) = x^4 - 4x^2 + \frac{5}{x+4}$.

> f := x -> x^4-4*x^2+5/(x+4);


$$f := x \rightarrow x^4 - 4x^2 + \frac{5}{x+4}$$

Untuk menggambar grafik fungsi dapat digunakan perintah **plot** dengan argumen rumus atau nama fungsi dan domain (nilai-nilai) variabel bebas. Berikut diberikan beberapa contoh bagaimana menggambar grafik fungsi dengan Maple.


> plot(9-x^3,x=-4..4); # rumus fungsi langsung ditulis



> plot(f(x), x=-4..4); # nama fungsi dan variabel




> `plot(f(x),x=-4..4,y=-4..4); # jangkauan x & y dibatasi`



Grafik beberapa fungsi juga dapat digambar dengan sebuah perintah plot.

> `plot({x,x^2,x^3,x^4},x=-1..1);`



2. Pendekatan Geometris terhadap Limit


Setelah melihat kembali beberapa konsep dasar, sekarang akan dinjau konsep limit dari sudut pandang geometris. Konsep limit mungkin terlihat sulit dipahami, namun ide tentang limit amatlah sederhana jika ditinjau secara geometris. Berikut diberikan sebuah diagram yang memperlihatkan kurva sebuah fungsi dan titik-titik yang mendekati sebuah titik dari kiri dan kanan.

> `f := x -> 3+(x-2)*cos((x-2)); a:=2: kiri:=-1: kanan:=5:`

$$f := x \rightarrow 3 + (x - 2) \cos(x - 2)$$

```
> display(plot(f(x), x=kiri..kanan, color=green),
plot([[a,0],[a,f(a)]],[[0,f(a)],[a,f(a)]]),x=kiri..kanan,
linestyle=3,color=gold,thickness=2),
plot([[a-1/n,f(a-1/n)] $n=1..20],x=kiri..kanan,
style=point, symbol=circle, color = red),
plot([[a+1/n,f(a+1/n)]$n=1..20],x=kiri..kanan,
```

```
style=point, symbol=circle, color = blue));
```




Misalkan $(a, f(a))$ adalah sebuah titik pada kurva $y = f(x)$. Pada diagram di atas diperlihatkan titik-titik pada kurva tersebut apabila x mendekati a dari kiri dan dari kanan garis $x = a$. Dengan mengamati nilai y pada saat x hampir mendekati a dari kiri, Anda dapat menebak limit kiri. Secara serupa, dengan mengamati nilai y pada saat x hampir mendekati a dari kanan, Anda dapat menebak limit kanan. Jika titik-titik tersebut bertemu di titik $(a, f(a))$, maka limit f di a ada dan sama dengan $f(a)$.

Berikut diperlihatkan sebuah contoh di mana limit kiri dan limit kanan sebuah fungsi tidak sama.

```
>f:=x->Heaviside(x-1)-Heaviside(1-x); a:=1: kiri:=-2: kanan:=4:
```

```
f := x → Heaviside(x - 1) - Heaviside(1 - x)
```

```
> display(plot(f(x),x=kiri..kanan, color=green),
plot([[a,0],[a,f(a)]],[[0,f(a)],[a,f(a)]],x=kiri..kanan,
linestyle=3,color=gold, thickness=2),
plot([[a-1/n,f(a-1/n)]$n=1..20],x=kiri..kanan,
style=point, symbol=circle, color = red),
plot([[a+1/n,f(a+1/n)]$n=1..20], x=kiri..kanan,
style=point, symbol=circle, color = blue));
```



Pada kasus ini, fungsinya tidak kontinu, karena mempunyai lompatan di $x = 1$. Titik-titik di sebelah kiri $x=1$ mendekati nilai $y = -1$ apabila nilai x mendekati 1. Sebaliknya, titik-titik di sebelah kanan $x=1$ mendekati nilai $y = 1$ apabila nilai x mendekati 1. Dalam hal dikatakan fungsi tersebut tidak mempunyai limit di $x = 1$.

3. Pendekatan Numeris terhadap Limit

Setelah melihat apa yang terjadi secara geometris, sekarang kita akan meninjau fungsi yang sama secara numeris. Kita akan menghitung nilai-nilai $f(x)$ untuk x mendekati a dari kiri ($x < a$) dan dari kanan

($a < x$).

> $f := x \rightarrow 3+(x-2)*\cos((x-2)); a:=2:$

$$f := x \rightarrow 3 + (x - 2) \cos(x - 2)$$

Tabel nilai-nilai $f(x)$ untuk x mendekati $a = 2$ dari kiri:

>array([seq([evalf(a-10^(-k),15),evalf(f(a-10^(-k)),15)],k=1..12)]);

1.900000000000000	2.90049958347220
1.990000000000000	2.99000049999583
1.999000000000000	2.99900000050000
1.999900000000000	2.99990000000050
1.999990000000000	2.99999000000000
1.999999000000000	2.99999900000000
1.999999900000000	2.99999990000000
1.999999990000000	2.99999999000000
1.999999999000000	2.99999999900000
1.999999999900000	2.99999999990000
1.999999999990000	2.99999999999000
1.999999999999000	2.99999999999900

Tabel nilai-nilai $f(x)$ untuk x mendekati $a = 2$ dari kanan:

>array([seq([evalf(a+10^(-k),15),evalf(f(a+10^(-k)),15)],k=1..12)]);

2.100000000000000	3.09950041652780
2.010000000000000	3.00999950000417
2.001000000000000	3.00099999500000
2.000100000000000	3.00009999999500
2.000010000000000	3.00001000000000
2.000001000000000	3.00000100000000
2.000000100000000	3.00000010000000
2.000000010000000	3.00000001000000
2.000000001000000	3.00000000100000
2.000000000100000	3.00000000010000
2.000000000010000	3.00000000001000
2.000000000001000	3.00000000000100

Dari kedua tabel di atas, Anda dapat melihat bahwa nilai-nilai y semakin mendekati 3 baik untuk x yang mendekati 2 dari kiri maupun kanan. Dari kenyataan ini Anda dapat menebak bahwa limit f di $x = 2$ ada, yakni 3.

Sekarang cobalah buat tabel nilai-nilai fungsi seperti di atas untuk kasus fungsi berikut ini dengan $a = 3$.

> $f := x \rightarrow \text{piecewise}(x < 3, -5 + x, x \leq 6, 20 - x^2, x < 10, 2 * x + 1.1); a := 6:$

$$f := x \rightarrow \text{piecewise}(x < 3, -5 + x, x \leq 6, 20 - x^2, x < 10, 2x + 1.1)$$

Dalam kasus ini Anda akan mendapatkan bahwa limit kiri dan limit kanan tidak sama. Nilai limit kiri adalah -2, sedangkan nilai limit kanan adalah 11. Apabila nilai limit kiri tidak sama dengan nilai limit kanan dikatakan limitnya tidak ada.

4. Pendekatan Simbolik terhadap Limit


Maple dapat digunakan untuk menghitung limit fungsi secara simbolik. Perintah **Limit** dengan "L" besar dapat digunakan untuk menghitung limit suatu fungsi seperti notasi baku dalam matematika. Pemakaiannya sangat mudah, tuliskan rumus atau nama fungsi dan nilai di mana limit hendak dihitung.

Untuk menghitung limit kanan atau kiri gunakan kata-kata "right" atau "left" sebagai argumen ketiga. Perhatikan contoh di bawah ini.

> f := x -> 3+(x-2)*cos((x-2)); a := 2;

$$f := x \rightarrow 3 + (x - 2) \cos(x - 2)$$

> plot([f(x),[[2,0],[2,f(2)]],[[2,f(2)],[0,f(2)]],x=0..5);



> Limit(f(x),x=a, left): % = value(%);

$$\lim_{x \rightarrow 2^-} 3 + (x - 2) \cos(x - 2) = 3$$

> Limit(f(x),x=a, right): % = value(%);

$$\lim_{x \rightarrow 2^+} 3 + (x - 2) \cos(x - 2) = 3$$

> Limit(f(x),x=a): % = value(%);

$$\lim_{x \rightarrow 2} 3 + (x - 2) \cos(x - 2) = 3$$

B. Definisi Derivatif

Pada bagian ini kita akan mengembangkan konsep derivatif, secara geometris dan aljabaris. Kita akan melihat tali-tali busur yang melewati dua buah titik pada kurva sebuah fungsi dan melihat bagaimana sekumpulan tali busur yang melewati sebuah titik tetap mengarah kepada definisi garis singgung. Akhirnya, kita akan menghitung derivatif sebuah fungsi dengan menggunakan definisi derivatif dan selisih terbagi. Seperti biasanya, jalankan program Maple (jika belum), kemudian tuliskan perintah awal sebagai berikut.

> restart; with(plots):

1. Tali Busur

Suatu tali busur adalah sebuah garis lurus yang melalui dua buah titik pada sebuah kurva suatu fungsi. Berikut ditunjukkan pemakaian Maple untuk menggambar tali-busur pada sebuah kurva suatu fungsi. Mula-mula kita definisikan sebuah fungsi dan dua buah nilai x , yakni $x = a$ dan $x = b$:

> f := x -> 20-10*cos(x)+(x^2)*sin(x); a := 1; b:=3;

$$f := x \rightarrow 20 - 10 \cos(x) + x^2 \sin(x)$$

$$a := 1$$

$$b := 3$$

Selanjutnya kita hitung gradien garis (tali-busur) yang melalui titik $(a, f(a))$ dan $(b, f(b))$ dengan menggunakan rumus biasa:

> m := evalf(f(b)-f(a))/(b-a);

$$m := 7.865778560$$


Berikutnya kita definisikan tali busur sebagai fungsi $L(x)$ dengan menyelesaikan persamaan

$y - f(a) = m(x - a)$. Akhirnya kita gambar kurva $y = f(x)$ dan $y = L(x)$.

> $L := x \rightarrow f(a) + m(x - a)$;

$$L := x \rightarrow f(a) + m(x - a)$$

> `display(plot({f(x),L(x)},x=0..8,thickness=2),
plot([[a,0],[a,f(a)]],[[b,0],[b,f(b)]],x=0..8,color=blue));`



Anda dapat mencoba menggambar tali-tali busur yang lain pada kurva tersebut.

2. Garis Singgung


Apabila tali-busur melalui dua titik berlainan pada kurva sebuah fungsi, garis singgung melalui tepat sebuah titik pada kurva suatu fungsi. Sekalipun ada tak berhingga garis yang melalui titik tersebut, namun hanya ada tepat satu garis singgung yang melalui suatu titik pada sebuah kurva. Garis-garis singgung memberikan arah perjalanan sebuah kurva. Salah satu tujuan utama kalkulus adalah mencari gradien garis-garis singgung karena mereka memberikan informasi tentang arah kurva pada setiap titik.

Tuliskan perintah-perintah Maple berikut ini pada layar Maple untuk memperlihatkan sebuah animasi garis-garis singgung pada kurva sebuah fungsi di titik-titik yang berlainan. Klik gambar yang dihasilkan, kemudian klik tombol **"Play"** (berupa simbol bukan tulisan).

> $f := x \rightarrow x^3 - 4x^2 + 2x + 2$;

> $T := (x,a) \rightarrow f(a) + (x-a) * D(f)(a)$;

> `display(plot(f(x),x=-2..4,y=-4..8,thickness=3,color=red),
animate(T(x,t),x=-2..4,t=-1.1..3.9,color=blue,frames=100));`



3. Keluarga Tali Busur

Salah satu tujuan utama kalkulus adalah mencari gradien garis-garis singgung kurva. Akan tetapi, tidaklah mungkin (tanpa mengetahui metode kalkulus) menghitung gradien sebuah garis singgung dengan hanya

mengetahui sebuah titik tinggung (karena rumus $m = \frac{y_2 - y_1}{x_2 - x_1}$ memerlukan dua buah titik berlainan). Di sisi lain, cukuplah mudah menghitung gradien tali busur. Kita akan menggunakan sesuatu yang mudah didapat untuk memperoleh sesuatu yang sulit didapat. Pendekatan yang kita gunakan adalah menaksir gradien garis singgung dengan menghitung gradien tali-tali busur yang semakin mendekati garis singgung.

Sebuah diagram akan memperlihatkan gagasan tersebut. Kita akan menggambar suatu keluarga tali busur yang melalui sebuah titik tetap $(c, f(c))$ dan titik lain berbentuk $(c + \frac{2}{k}, f(c + \frac{2}{k}))$ secara bersamaan dan kita ingin mencari garis singgung di titik tersebut. Kedua titik tersebut akan semakin dekat semakin besar nilai k . Berikut adalah perintah-perintah Maple yang dapat digunakan.

```
> c := 2;
```

$$c := 2$$

```
> f := x -> 20 - 10*cos(x) + (x^2)*sin(x);
```


$$f := x \rightarrow 20 - 10 \cos(x) + x^2 \sin(x)$$

```
> TB := (a,b) -> f(a) + ((f(b)-f(a))/(b-a))*(x-a);
```

$$TB := (a, b) \rightarrow f(a) + \frac{(f(b) - f(a))(x - a)}{b - a}$$

```
> display(plot(f(x),x=0..8,y=-15..40,color=red,thickness=2),
```

```
plot({TB(c,c+2/k)$k=1..8},x=0..8,y=-15..40,color=blue), pointplot([c,f(c)]));
```




Terlihat bahwa keluarga tali-busur semakin mendekati garis singgung. Akibatnya, gradien tali-tali busur juga mendekati gradien garis singgung. Berikut adalah pandangan yang lebih dekat hal yang sama untuk memperlihatkan apa yang terjadi secara lebih jelas.

```
> display(pointplot([c,f(c)]),plot(f(x),x=(c-1)..(c+1),y=0..40,
```

```
thickness=2),plot({TB(c,c+2/k)$k=2..8},x=(c-1)..(c+1),
```

```
y=0..40,color=blue));
```



4. Definisi Derivatif

Sekarang kita akan melihat gradien garis singgung dan derivatif dari sudut pandang aljabar.

Misalkan kita punya fungsi sebagai berikut.

```
> f := x -> x^3 + 1;
```

$$f := x \rightarrow x^3 + 1$$

Pertama-tama, kita akan mencari gradien garis singgung kurva di $x = 2$. Untuk ini, kita menghitung gradien tali busur yang melalui titik $(2, f(2))$ dan $(2 + h, f(2 + h))$, menyederhanakan ekspresinya, dan mengambil limitnya untuk h mendekati 0, serta mengubahnya ke nilai (bilangan) desimal. Gradien tali busur yang melalui titik $(2, f(2))$ dan $(2 + h, f(2 + h))$ adalah:

```
> (f(2+h)-f(2))/h;
```

$$\frac{(2+h)^3 - 8}{h}$$

Sederhanakan ekspresi tersebut:

```
> simplify(%);
```

$$12 + 6h + h^2$$

Ambil limitnya untuk h mendekati 0:

```
> limit(%,h=0);
```

$$12$$

Hitung nilai tersebut sebagai bilangan desimal.

```
> gradien_garis_singgung:=evalf(%);
```

$$\text{gradien_garis_singgung} := 12.$$

Apabila langkah-langkah di atas diulangi dengan membiarkan nilai x berupa variabel, maka akan diperoleh gradien garis singgung kurva $y = f(x)$ di titik $(x, f(x))$:

```
> (f(x+h) - f(x))/h;
```

$$\frac{(x+h)^3 - x^3}{h}$$

```
> simplify();
```

$$3x^2 + 3xh + h^2$$

```
> limit(%, h=0);
```

$$3x^2$$

C. Lebih Jauh tentang Turunan (Derivatif) Fungsi

Setelah mempelajari konsep limit dan turunan, sekarang kita akan membahas berbagai cara untuk menghitung turunan fungsi, bagaimana menghasilkan tabel turunan untuk melihat kecenderungan, menghitung turunan tingkat tinggi, dan membuat garis-garis singgung kurva. Seperti biasanya, jika Anda belum menjalankan Maple, jalankan Maple dan tulis perintah awal sebagai berikut.

```
> restart; with(linalg);
```

1. Sekali lagi tentang Derivatif

Pada bagian sebelumnya kita telah belajar bagaimana mencari derivatif melalui "jalan panjang". Sekarang kita akan menggunakan perintah Maple untuk menghitung derivatif suatu fungsi. Perintah yang digunakan adalah **diff** atau dengan operator **D**.

Perhatikan contoh di bawah ini. Mula-mula kita definisikan fungsinya seperti biasa, misalnya:

> f := x -> sqrt(1+x^2);

$$f := x \rightarrow \sqrt{1+x^2}$$

Untuk menghitung derivatif dengan menggunakan perintah diff diperlukan dua buah parameter, yakni fungsi dan variabel terhadap mana derivatif dilakukan. Untuk menghitung turunan fungsi di atas dapat digunakan perintah Maple sebagai berikut.

> diff(f(x),x);

$$\frac{x}{\sqrt{1+x^2}}$$

Hasil yang sama dapat diperoleh dengan menggunakan operator derivatif **D** seolah-olah operator **D** ini bertindak sebagai fungsi sehingga cara pemakaiannya sangatlah mudah. Perhatikan seksama contoh perhitungan derivatif dengan menggunakan operator **D** berikut ini.

> D(f)(x);

$$\frac{x}{\sqrt{1+x^2}}$$

Masih terdapat cara lain untuk menghitung derivatif suatu fungsi pada Maple, yakni dengan menggunakan perintah **Diff**. Akan tetapi, perintah **Diff**, tidak secara langsung menghasilkan ekspresi hasil turunan, melainkan notasi seperti dalam matematika, yakni $\frac{df}{dx}$, dan untuk menghasilkan ekspresi turunannya digunakan perintah **value**. Perhatikan contoh sebagai berikut untuk menghitung turunan fungsi di atas.

> Diff(f(x),x); %=value(%);

$$\frac{\partial}{\partial x} \sqrt{1+x^2}$$
$$\frac{\partial}{\partial x} \sqrt{1+x^2} = \frac{x}{\sqrt{1+x^2}}$$

Jadi, Anda dapat menyimpulkan bahwa **diff(f(x),x)=value(Diff(f(x),x))**.

2. Tabel Derivatif

Selain menghitung derivatif sebuah fungsi, kita dapat menggunakan sebuah perintah Maple untuk menghitung tabel turunan beberapa fungsi hanya dengan menggunakan sebuah perintah Maple. Dengan perkataan lain, kita dapat menghasilkan matriks derivatif suatu matriks fungsi. Cara ini dapat memudahkan kita melihat pola dan kecenderungan suatu derivatif. Berikut adalah contoh menghitung matriks derivatif. Perhatikan perintah-perintah Maple yang dipakai.

> A:=matrix(2,2,[sin(x), x^2+x+3, exp(x), cos(x^2)]);

$$A := \begin{bmatrix} \sin(x) & x^2 + x + 3 \\ e^x & \cos(x^2) \end{bmatrix}$$

> Turunan_A:=map(diff,A,x);

$$\text{Turunan_A} := \begin{bmatrix} \cos(x) & 2x + 1 \\ e^x & -2 \sin(x^2) \cdot x \end{bmatrix}$$

Untuk menghitung turunan barisan fungsi lebih mudah lagi, seperti contoh di bawah ini.

> matrix(2,11,['f(x)',seq(x^k,k=1..10),D('f(x)'),seq(diff(x^k,x),
k= 1..10)]);

$$\begin{bmatrix} f(x) & x & x^2 & x^3 & x^4 & x^5 & x^6 & x^7 & x^8 & x^9 & x^{10} \\ D(f(x)) & 1 & 2x & 3x^2 & 4x^3 & 5x^4 & 6x^5 & 7x^6 & 8x^7 & 9x^8 & 10x^9 \end{bmatrix}$$

> matrix(1,11,['f(x)',seq(x^k,k=1..10)]);

$$[f(x) \quad x \quad x^2 \quad x^3 \quad x^4 \quad x^5 \quad x^6 \quad x^7 \quad x^8 \quad x^9 \quad x^{10}]$$

Perhatikan pola pada tabel turunan berikut ini.

> matrix(2,11,['f(x)',seq(x^k,k=-5..4),D('f(x)'),seq(diff(x^k,x),k=-5..4)]);

$$\begin{bmatrix} f(x) & \frac{1}{x^5} & \frac{1}{x^4} & \frac{1}{x^3} & \frac{1}{x^2} & \frac{1}{x} & 1 & x & x^2 & x^3 & x^4 \\ D(f(x)) & -5 \frac{1}{x^6} & -4 \frac{1}{x^5} & -3 \frac{1}{x^4} & -2 \frac{1}{x^3} & -\frac{1}{x^2} & 0 & 1 & 2x & 3x^2 & 4x^3 \end{bmatrix}$$

Berikut adalah tabel turunan fungsi-fungsi $\sin(x)$, $\sin(2x)$, $\sin(3x)$, $\sin(4x)$, dan $\sin(5x)$. Dapatkan Anda melihat pola yang ada? Perhatikan pula perintah-perintah Maple yang dipakai, berbeda dengan perintah-perintah sebelumnya.

> transpose(array([seq([sin(k*x),diff(sin(k*x),x)],k=1..5)]));

$$\begin{bmatrix} \sin(x) & \sin(2x) & \sin(3x) & \sin(4x) & \sin(5x) \\ \cos(x) & 2 \cos(2x) & 3 \cos(3x) & 4 \cos(4x) & 5 \cos(5x) \end{bmatrix}$$

Cobalah Anda berlatih membuat tabel atau matriks turunan baran atau matriks fungsi, baik yang beraturan maupun bebas (sebarang fungsi).

3. Menghitung Nilai Derivatif

Oleh karena derivatif suatu fungsi adalah fungsi, maka kita dapat menghitung nilai derivatif di berbagai nilai x . Dengan kata lain, kita dapat mengganti nilai x pada $f'(x)$, malnya untuk menghitung nilai $f'(a)$. Nilai ini tidak lain adalah gradien gar singgung di titik $(a, f(x))$. Dalam Maple hal ini dapat dilakukan dengan setidaknya menggunakan tiga cara.

Pertama, dengan menggunakan perintah `diff` kita masih memerlukan perintah kedua, yakni `eval` untuk mendapatkan hasilnya. **Kedua**, dengan menggunakan operator **D**, penulannya menjadi lebih singkat. **Ketiga**, dengan menghitung nilai fungsi turunan. Bandingkan ketiga cara sebagai berikut. Cara manakah yang paling nyaman dan mudah dipakai?

> f := x -> x^4 + x^3 - 5*x^2 + 60;

$$f := x \rightarrow x^4 + x^3 - 5x^2 + 60$$

> diff(f(x), x); eval(%, x=3);

$$4x^3 + 3x^2 - 10x$$

$$105$$

> D(f)(3);

$$105$$

> g:=x -> D(f)(x); g(3);

$$g := x \rightarrow 4x^3 + 3x^2 - 10x$$

$$105$$

Dapatkan Anda menggunakan cara lain untuk menghitung nilai turunan suatu fungsi di suatu titik?

4. Garis Singgung

Seperti sudah Anda ketahui, nilai turunan $f'(x)$ di $x = a$ merupakan gradien garis singgung kurva di titik $(a, f(a))$. Persamaan garis singgung tersebut dapat dinyatakan sebagai $y = f(a) + f'(a)(x-a)$. Dalam contoh berikut kita definisikan fungsi f dan kita tentukan persamaan garis singgungnya di titik $a = 2$. Fungsi T menyatakan fungsi garis singgung f . Grafik kedua fungsi kita gambar pada sumbu koordinat yang sama. Perhatikan langkah-langkahnya.

```
> a := 2;
```

$$a := 2$$


```
> f := x -> 2 + sin(x^2)/(x-1);
```

$$f := x \rightarrow 2 + \frac{\sin(x^2)}{x-1}$$

```
> T := x -> f(a) + D(f)(a) * (x-a);
```

$$T := x \rightarrow f(a) + D(f)(a) (x - a)$$

```
> plot( {f(x),T(x)},x=-2..5,y=-10..10,dcont=true);
```



Seperti pada pembahasan dan contoh sebelumnya, jika Anda perhatikan, untuk mendefinisikan fungsi turunan kita menggunakan operator **D**, bukan perintah Maple **diff** atau **Diff**.

5. Derivatif Tingkat Tinggi

Dengan menggunakan perintah **diff** atau operator **D** pada Maple kita dapat menghitung derivatif tingkat tinggi, misalnya turunan kedua, ketiga, keempat, dan seterusnya. Perhatikan contoh-contoh berikut ini.

```
> f := x -> x^3+x^2+x+7;
```

$$f := x \rightarrow x^3 + x^2 + x + 7$$

```
> diff(f(x),x); diff(f(x),x$2); diff(f(x),x$3);
```

$$3x^2 + 2x + 1$$

$$6x + 2$$

$$6$$

```
> D(f)(x); (D@@2)(f)(x); (D@@3)(f)(x);
```

$$3x^2 + 2x + 1$$

$$6x + 2$$

$$6$$

Seperti Anda lihat, jika kita gunakan perintah **diff**, simbol dollar (\$) digunakan di belakang variabel penurunan untuk menyatakan derajat turunan yang diinginkan. Jika Anda menggunakan operator **D**,

simbol **at (@)** dipakai di belakang operator tersebut untuk menyatakan derajat turunan yang diinginkan.

Kita juga dapat menghasilkan tabel berbagai tingkat turunan sebuah fungsi, seperti contoh berikut ini.

```
> f := x -> sin(7*x);
```

$$f := x \rightarrow \sin(7x)$$

```
> transpose(array([seq([k,(D@@k)(f)(x)],k=0..4)]));
```

$$\begin{bmatrix} 0 & 1 & 2 & 3 & 4 \\ \sin(7x) & 7 \cos(7x) & -49 \sin(7x) & -343 \cos(7x) & 2401 \sin(7x) \end{bmatrix}$$

Tabel tersebut merupakan tabel turunan ke- k fungsi $f(x) = \sin(7x)$. Perhatikan, untuk $k = 0$ hasilnya adalah fungsi itu sendiri. Berikut adalah tabel turunan fungsi lain.

```
> f := x -> sqrt(x);
```

$$f := \text{sqrt}$$

```
> transpose(array([seq([k,(D@@k)(f)(x)],k=0..4)]));
```


$$\begin{bmatrix} 0 & 1 & 2 & 3 & 4 \\ \sqrt{x} & \frac{1}{2} \frac{1}{\sqrt{x}} & -\frac{1}{4} \frac{1}{x^{(3/2)}} & \frac{3}{8} \frac{1}{x^{(5/2)}} & -\frac{15}{16} \frac{1}{x^{(7/2)}} \end{bmatrix}$$

6. Grafik Fungsi & Fungsi Turunannya

Sekali lagi, derivatif sebuah fungsi juga merupakan fungsi. Hal ini berlaku untuk turunan kedua, ketiga, dan yang lebih tinggi. Fungsi-fungsi $f(x)$, $f'(x)$, $f''(x)$, ... dapat kita gambarkan bersama-sama pada sebuah sumbu koordinat. Hal ini dapat dilakukan dengan mudah menggunakan Maple. Perhatikan contoh di bawah ini.

```
> f := x->sqrt(1+x^2); plot({(D@@k)(f)(x)$k=0..4},x=-2..2);
```

$$f := x \rightarrow \sqrt{x^2 + 1}$$



Perhatikan dengan cermat, kita cukup menggunakan sekali perintah **plot** untuk menggambar kurva sebuah fungsi fungsi-fungsi turunannya.

Anda perlu melakukan inspeksi secara hati-hati untuk mengetahui mana kurva suatu fungsi. Berikut adalah beberapa petunjuk yang dapat Anda gunakan:

1. Apabila $f(x)$ mulus dan mempunyai maksimum atau minimum lokal, $f'(x)$ memotong sumbu- x .
2. Apabila $f(x)$ naik, $f'(x)$ di atas sumbu- x .
3. Apabila $f(x)$ turun, $f'(x)$ di bawah sumbu- x .
4. Apabila $f(x)$ cekung ke atas, $f'(x)$ naik, dan $f''(x)$ di atas sumbu- x .
5. Apabila $f(x)$ cekung ke bawah, $f'(x)$ decreasing, dan $f''(x)$ di bawah sumbu- x .

D. Bentuk Kurva

Pada bagian ini Anda akan melakukan eksplorasi hubungan antara turunan pertama dan kedua suatu fungsi dan bentuk kurvanya. Penelaahan konsep-konsep ini dari sudut pandang aljabar dan geometrik akan memperlihatkan kepada Anda bagaimana nilai turunan mempengaruhi bentuk kurva dan sebaliknya mengenalinya dari kurva fungsi.

> `restart; with(plots):`

1. Pendekatan Simbolik

Derivatif sebuah fungsi memberikan banyak informasi tentang bentuk kurva fungsi tersebut. Turunan pertama memberitahukan apakah kurvanya naik atau turun. Apabila $f'(x) > 0$, maka fungsinya naik, dan apabila $f'(x) < 0$, maka fungsinya turun. Dengan Maple kita dapat mencari interval di mana kurva suatu fungsi naik atau turun. Perhatikan contoh berikut ini untuk mengetahui interval di mana kurva fungsi naik.

> `f := x -> x^3 - x^2 - x + 4;`


$$f := x \rightarrow x^3 - x^2 - x + 4$$

> `Naik := solve(D(f)(x)>0,{x});`

$$Naik := \left\{ x < \frac{-1}{3} \right\}, \{ 1 < x \}$$

Untuk meyakinkan hal itu, kita dapat menggambar kurva fungsinya.

> `plot(f(x),x=-1.5..2);`



Secara serupa, turunan kedua menginformasikan apakah kurva fungsinya cekung ke atas atau ke bawah. Turunan kedua positif ($f''(x) > 0$) menunjukkan kurva cekung ke atas, sedangkan turunan kedua negatif menunjukkan kurva cekung ke bawah. Perhatikan cara menentukan interval di mana kurva cekung ke atas dengan Maple di bawah ini.

> `Cekung_atas := solve((D@@2)(f)(x)>0,{x});`

$$Cekung_atas := \left\{ \frac{1}{3} < x \right\}$$

Bandingkan hasil numerik ini dengan kurva di atas.

Perhatikan juga cara Maple menuliskan interval jawaban. Maple memberikan notasi yang berlainan dengan notasi matematika biasa untuk menyatakan interval. Beberapa contoh adalah sebagai berikut.

Output Maple	: {x < 5}, {9 < x}; {2 < x, x < 3}
Notasi Himpunan	: {x : x < 5 or 9 < x}; {x : 2 < x < 3}
Notasi Interval	: (-inf, 5) U (9, inf); (2, 3)

Dalam beberapa masalah Maple memberikan jawaban yang benar namun mungkin agak rumit kelihatannya. Kerumitan ekspresi ini dapat diatasi atau dihindari dengan menggunakan perintah `evalf(%)` yang langsung digunakan setelah perintah terakhir. Perhatikan contoh-contoh di bawah ini.

> `f := x -> x^4 - 16*x^3 + 89*x^2 - 200*x + 140;`

$$f := x \rightarrow x^4 - 16x^3 + 89x^2 - 200x + 140$$

> Naik := solve(D(f)(x)>0,{x}); evalf(%);

$$\text{Naik} := \{x < 4, 4 - \frac{1}{2}\sqrt{14} < x\}, \{4 + \frac{1}{2}\sqrt{14} < x\}$$

$$\{2.129171306 < x, x < 4.\}, \{5.870828694 < x\}$$

> Cekung_atas := solve((D@@2)(f)(x)>0,{x}); evalf(%);

$$\text{Cekung_atas} := \{x < 4 - \frac{1}{6}\sqrt{42}\}, \{4 + \frac{1}{6}\sqrt{42} < x\}$$


$$\{x < 2.919876550\}, \{5.080123450 < x\}$$

Seperti Anda lihat hasil sebelum perintah **evalf** dalam bentuk ekspresi yang cukup rumit, namun hasil setelah menggunakan perintah tersebut sangat sederhana, karena dalam bentuk pechan desimal.


2. Pendekatan Geometrik

Sekarang kita akan melakukan eksplorasi sifat-sifat turunan dari sudut pandang geometris. Format penyajian kali ini agak berbeda, yakni mula-mula kita gambar kurva fungsi-fungsi berikut ini kemudian mencetak (menampilkan) gambar-gambar tersebut.


> plot(x^3-x^2-x+4,x=-1.6.. 2);




> plot(x^6-5*x^4-12*x^3-x^2+15*x+18,x=-2..3);



> plot(x/(x^2+1),x=-3..3);



> plot((9-x)/(1+x^2),x=-5..5);



Berikut diberikan sebuah blok kode program (prosedur) Maple, yang diberi nama **sifat_kurva**, untuk mendefinisikan suatu fungsi untuk menggambar kurva sebuah fungsi dan daerah-daerah di mana kurvanya naik, turun, cekung ke atas, dan cekung ke bawah.

```


> with(plots):
sifat_kurva := proc(f1,a,b)
local box,i,n,x1,x2,xmid,delta,y1,y2,A,B,m,M,gradien,cekung:
n:=100; delta:=(b-a)/n; x2:=a;
M:=maximize(f(x),x=a..b); m:=minimize(f(x),x=a..b);
for i from 1 to n do
    x1:=evalf(x2); y1:=evalf(f(x1));
    x2:=evalf(a+i*delta); y2:=evalf(f(x2)); xmid:=x1+delta/2;
    gradien:=evalf(subs(x=xmid,diff(f1,x)));
    cekung:=evalf(subs(x=xmid,diff(f1,x$2)));
    if(gradien>0)
        then A[i]:=plot(f1,x=x1..x2,color=blue,thickness=4);
        else A[i]:=plot(f1,x=x1..x2,color=red,thickness=2);
    fi;
    if(cekung>0) then
    B[i]:=polygonplot([[x1,M],[x1,y1],[x2,y2],[x2,M]],
        color=green,style=patchnogrid);
    else
    B[i]:=polygonplot( [[x1,m],[x1,y1],[x2,y2],[x2,m]],
        color=brown,style=patchnogrid);
    fi;
od;
display({seq(A[i],i=1..n),seq(B[i],i=1..n)});
end:

```

Berikut adalah contoh pemakaian prosedur di atas. Perhatikan gambar yang dihasilkan dan pelajari sifat-sifat kurva dengan menggunakan gambar tersebut. Cobalah Anda cek hasil tersebut dengan menghitung turunan pertama dan kedua fungsi tersebut.

> f:=x->x^3-x^2-x+4; sifat_kurva(f(x),-1.6,2);

$$f := x \rightarrow x^3 - x^2 - x + 4$$



Ketebalan (dan warna) memperlihatkan sifat-sifat setiap bagian kurva (turun, naik, cekung ke atas, atau cekung ke bawah).

Warna/ketebalan kurva:

Biru (tebal) : kurva naik (turunan pertama positif)
Merah (tipis) : kurva turun (turunan pertama negatif)


Warna daerah di atas/dibawah kurva:

Coklat (di bawah kurva) : kurva cekung ke atas

Hijau (di atas kurva) : kurva cekung ke bawah


Perhatikan, maksimum lokal terjadi pada titik pergantian warna/ketebalan kurva dari biru (tebal) ke merah (tipis). Minimum lokal terjadi pada titik pergantian warna/ketebalan dari merah (tipis) ke biru (tebal). Titik belok terjadi pada bergantian daerah berwarna dari hijau ke coklat ke hijau atau sebaliknya. Amati dan perhatikan contoh-contoh berikut ini.

> f := x -> x^6-5*x^4-12*x^3-x^2+15*x+18; sifat_kurva(f(x),-2,3);
 $f := x \rightarrow x^6 - 5x^4 - 12x^3 - x^2 + 15x + 18$




> f := x -> x/(x^2+1); sifat_kurva(f(x),-3,3);

$$f := x \rightarrow \frac{x}{x^2 + 1}$$



> f := x -> x^2+1/(x^2+1); sifat_kurva(f(x),-3,3);

$$f := x \rightarrow x^2 + \frac{1}{x^2 + 1}$$



Bagian II: Kalkulus Integral

A. Definisi Integral

Pada bagian ini kita akan belajar mengenai luas daerah di bawah urva, bagaimana jumlah Riemann konvergen ke luas daerah tersebut, bagaimana pemilihan batas kiri, kanan, titik tengah sub-subinterval tidak mempunyai efek. Selanjutnya kita akan beranjak melalui prosedur-empat langkah untuk menghitung jumlah Riemann. Sebelum Anda mencoba perintah-perintah Maple dalam materi ini, tuliskan perintah-perintah awal Maple berikut.

```
> restart; with(plots): with(student):
```

1. Daerah Positif & Negatif

Pertama-tama, Anda ingat konsep luas di bawah kurva dan lihat bagaimana luas tersebut dapat bernilai positif dan negatif. Untuk menyelidiki ide tersebut kita akan menuliskan sebuah prosedur (program) Maple yang bermanfaat untuk menghasilkan gambar kurva suatu fungsi dan daerah "di bawah" kurva pada interval-interval tertentu (yakni daerah yang dibatasi oleh sumbu- x dan kurva suatu fungsi dan dua buah garis tegak). *Sebaiknya Anda juga menuliskan prosedur di bawah ini sebelum Anda menggunakan perintah-perintah selanjutnya.*

```
> luas_kurva := proc(f,a,b)
  local box,i,n,x1,x2,y1,y2,A,delta:
  n:= 200; delta := (b-a)/n; x2 := a;
  for i from 1 to n do
    x1 := evalf(x2); y1 := evalf(f(x1));
    x2 := evalf(a+i*delta); y2 := evalf(f(x2));
    if( y1 > 0 )
    then A | i:=polygonplot([[x1,0],[x1,y1],[x2,y2],[x2,0]],
      color=coral,style=patchnogrid);
    else A | i:=polygonplot([[x1,0],[x1,y1],[x2,y2],[x2,0]],
      color=red, style=patchnogrid);
    fi; od;
  display([plot(f(x),x=a..b,color=blue,thickness=2,discont=true), seq(A | i,i=1..n)]);
end:
```

Untuk menggunakan prosedur Maple **luas_kurva** tersebut, Anda memerlukan masukan (argumen) berupa nama fungsi (harus didefinisikan terlebih dahulu, tidak dapat langsung dituliskan rumus fungsinya) dan kedua ujung suatu interval pada daerah asal fungsi. Perhatikan contoh di bawah ini.


```
> g := x -> 1+x/2+(x/3)*sin(x^2); a:=1; b:=3;
```

$$g := x \rightarrow 1 + \frac{1}{2}x + \frac{1}{3}x \sin(x^2)$$

$$a := 1$$

$$b := 3$$


```
> luas_kurva(g,a,b);
```



Pada contoh berikut ini daerah yang digambar memuat daerah positif (di atas sumbu- x) dan daerah negatif (di bawah sumbu- x).

> $h:=x \rightarrow x^2+3$; $luas_kurva(h,-2,2)$; $luas_kurva(-h,-2,2)$;

$$h := x \rightarrow x^2 + 3$$




Daerah antara kurva suatu fungsi dan sumbu- x pada suatu interval dapat terletak di bawah sekaligus di atas sumbu- x , seperti contoh di bawah ini.

> $f := x \rightarrow 10 + 6x - x^2$; $a := -3$; $b := 9$; $luas_kurva(f,a,b)$;

$$f := x \rightarrow 10 + 6x - x^2$$

$$a := -3$$

$$b := 9$$



Cobalah Anda menggambar daerah di bawah kurva fungsi-fungsi lain pada interval-interval yang Anda inginkan.

2. Fungsi-fungsi Terintegralkan

Sekalipun syarat suatu fungsi diferensiabel adalah kontinu dan kurvanya mulus, luas daerah fungsi dapat didefinisikan secara jelas sekalipun fungsinya tidak kontinu ataupun mulus kurvanya. Fungsi-fungsi demikian diistilahkan integrable (terintegralkan) karena beberapa alasan yang Anda ketahui dalam pembahasan berikut ini.

Daerah antara kurva dan sumbu- x tetap dapat digambar untuk fungsi-fungsi tak kontinu, fungsi sepasang-sepasang, dan fungsi kontinu yang tak diferensiabel di mana-mana.


> f:= x->piecewise(x<2,-1, x<6,2-x/2, x<8,.5*x^2-6*x+18,1);

$$f := x \rightarrow \text{piecewise}\left(x < 2, -1, x < 6, 2 - \frac{1}{2}x, x < 8, .5x^2 - 6x + 18, 1\right)$$

> a:=0; b:=10; luas_kurva(f,a,b);

a := 0

b := 10




> f := x->sqrt(abs(x^2-9))-2; a := -5; b := 5; luas_kurva(f,a,b);

$$f := x \rightarrow \sqrt{|x^2 - 9|} - 2$$

a := -5

b := 5



Cobalah Anda lakukan hal yang sama untuk fungsi-fungsi diskontinyu yang lain.

3. Jumlah Riemann & Konvergensi

Sekarang kita akan menguji suatu skema untuk menghitung luas di bawah kurva. Idennya adalah membagi interval pada sumbu- x menjadi sub-subinterval kecil yang berjarak sama, kemudian menggambar suatu persegi panjang pada setiap subinterval untuk menghampiri luas daerah pada setiap subinterval. Jumlah semua persegi panjang merupakan hampiran luas daerah di bawah kurva pada interval semula. Semakin banyak subinterval yang dipakai, hampiran yang diperoleh semakin mendekati luas sebenarnya.

Persegipanjang dan Luas

Perintah Maple **rightsum** dapat digunakan untuk menghitung jumlah luas persegi panjang pada suatu interval dengan subinterval sejumlah yang ditentukan dan fungsi tertentu. Perintah **rightbox** berguna untuk menggambar kurva fungsi dan persegi-persegipanjang pada sub-subinterval yang ditentukan. Perhatikan contoh di bawah ini.

> f:= x->x^3-19*x^2+99*x-20; a:=1; b:=13; n:=10;

$$f := x \rightarrow x^3 - 19x^2 + 99x - 20$$

a := 1


$b := 13$

$n := 10$

> `rightsum(f(x),x=a..b,n):` `Luas := evalf(%,20);`

Luas := 1428.9600000000000000

> `rightbox(f(x),x=a..b,n, title=cat(`Jumlah luas =`,
convert(evalf(Luas,10),string)));`



Jumlah luas ke sepuluh persegi panjang tersebut mendekati luas daerah di bawah kurva.

Konvergensi

Kita dapat melakukan perhitungan tersebut berkali-kali dengan mengubah nilai n sambil mengamati nilai-nilai luas yang diperoleh, apakah mengalami perubahan menuju ke suatu nilai tertentu. Perhatikan contoh di bawah ini.

> `for k from 1 to 10 do print(cat(`Luas untuk n=`,200*k), `adalah`,
evalf(rightsum(f(x),x=a..b,200*k),20)); od;`

Luas untuk n=200, adalah, 1313.7744000000000000

Luas untuk n=400, adalah, 1310.8836000000000000

Luas untuk n=600, adalah, 1309.9216000000000000

Luas untuk n=800, adalah, 1309.4409000000000000

Luas untuk n=1000, adalah, 1309.1525760000000000

Luas untuk n=1200, adalah, 1308.9604000000000000

Luas untuk n=1400, adalah, 1308.8231510204081632

Luas untuk n=1600, adalah, 1308.7202250000000000

Luas untuk n=1800, adalah, 1308.6401777777777778

Luas untuk n=2000, adalah, 1308.5761440000000000

Anda dapat melihat bahwa nilai-nilai luas semakin konvergen bersamaan dengan kenaikan nilai n dari 200 sampai 2000. Anda dapat melakukan hal yang sama untuk nilai-nilai n semakin bertambah besar, misalnya dari 200 sampai 2000000. Cobalah, namun Anda harus bersabar menunggu komputer menghitung!

4. Jumlah Kiri, Kanan & Tengah

Pada saat menggambar persegi-persegi panjang di atas kita menghitung tinggi masing-masing persegi panjang dengan menghitung nilai $f(x)$ di ujung kanan setiap subinterval. Cara ini bukanlah satu-satunya yang dapat Anda gunakan. Anda dapat menggunakan nilai $f(x)$ di ujung kiri (dengan perintah **leftbox**) atau tengah-tengah (dengan perintah **middlebox**) subinterval yang bersangkutan. Marilah kita lihat bagaimana kedua metode terakhir tampak berbeda dan memberikan hasil yang berbeda.

Perhatikan baik-baik secara cermat ketiga gambar berikut ini. Perhatikan pada setiap gambar di mana setiap persegi panjang bertemu. Anda akan melihat perbedaan ketiga gambar tersebut.

```
> f:= x -> x^3-19*x^2+99*x-20;
```


$$f := x \rightarrow x^3 - 19x^2 + 99x - 20$$

```
> a:=1; b:=13;
```


$$a := 1$$

$$b := 13$$


```
> leftbox(f(x),x=1..13,20);
```



```
> rightbox(f(x),x=1..13,20);
```



```
> middlebox(f(x),x=1..13,20);
```



Anda dapat menyaksikan bahwa untuk persegi-persegipanjang yang menggunakan tinggi ujung kiri atau kanan sub-subinterval, baik gambar maupun hampiran luas yang diperoleh berbeda. Pada akhirnya, apakah benar metode perhitungan tinggi persegi-panjang membuat hasilnya berbeda? Marilah kita uji luas kiri (dengan perintah **leftsum**) dan luas kanan (dengan perintah **rightsum**) dan selisih keduanya secara numerik, untuk nilai n berubah dari 200 sampai 3000.

> $f := x \rightarrow x^3 - 19x^2 + 99x - 20$; $a := 1$; $b := 13$;

$$f := x \rightarrow x^3 - 19x^2 + 99x - 20$$

$$a := 1$$

$$b := 13$$

> `array([seq([200*k,evalf(leftsum(f(x),x=a..b,200*k),8),
evalf(rightsum(f(x),x=a..b,200*k),8),
evalf(rightsum(f(x),x=a..b,200*k)-leftsum(f(x),x=a..b,
200*k),8)],k=1..15)]);`

200	1302.2544	1313.7744	11.5200
400	1305.1236	1310.8836	5.7600
600	1306.0816	1309.9216	3.8400
800	1306.5609	1309.4409	2.8800
1000	1306.8486	1309.1526	2.3040
1200	1307.0404	1308.9604	1.9200
1400	1307.1774	1308.8231	1.6457
1600	1307.2802	1308.7202	1.4400
1800	1307.3602	1308.6402	1.2800
2000	1307.4241	1308.5761	1.1520
2200	1307.4765	1308.5238	1.0473
2400	1307.5201	1308.4801	.9600
2600	1307.5570	1308.4432	.8862
2800	1307.5887	1308.4115	.8228
3000	1307.6161	1308.3841	.7680

Pada tabel di atas, kolom pertama adalah nilai-nilai n , kolom kedua jumlah kiri, kolom ketiga jumlah kanan, dan kolom keempat adalah selisih kedua jumlah. Sekalipun luas-luas kiri tidak sama dengan luas-luas kanan, namun keduanya tampak konvergen ke nilai yang sama. Hal ini juga terlihat dari selisih keduanya (kolom terakhir) yang semakin mendekati nol.

Bagaimanakah perbandingan jumlah kiri, kanan, dan tengah? Berikut adalah tabel perbandingan ketiganya untuk nilai-nilai n dari 200 sampai 3000.

```
> array([seq([200*k,evalf(leftsum(f(x),x=a..b,200*k),8),
              evalf(middlesum(f(x),x=a..b,200*k),8),
              evalf(rightsum(f(x),x=a..b,200*k),8)],k=1..15)]);
```

200	1302.2544	1307.9928	1313.7744
400	1305.1236	1307.9982	1310.8836
600	1306.0816	1307.9992	1309.9216
800	1306.5609	1307.9996	1309.4409
1000	1306.8486	1307.9998	1309.1526
1200	1307.0404	1307.9998	1308.9604
1400	1307.1774	1307.9998	1308.8231
1600	1307.2802	1307.9999	1308.7202
1800	1307.3602	1307.9999	1308.6402
2000	1307.4241	1307.9999	1308.5761
2200	1307.4765	1308.0000	1308.5238
2400	1307.5201	1308.0000	1308.4801
2600	1307.5570	1307.9999	1308.4432
2800	1307.5887	1308.0000	1308.4115
3000	1307.6161	1308.0000	1308.3841

Sekali lagi, meskipun ketiga kolom terakhir memiliki nilai-nilai yang berbeda, namun ketiganya tampak semakin menuju ke nilai yang sama. Jadi, Anda dapat menyimpulkan bahwa metode perhitungan tinggi persegi panjang pada setiap subinterval sebenarnya tidak berpengaruh terhadap jumlah total hampiran luas di bawah kurva yang dihasilkan, khususnya untuk nilai-nilai n semakin besar.

5. Menghitung Jumlah Riemann

Seperti sudah kita gunakan di atas, Maple menyediakan fungsi-fungsi yang sangat bermanfaat untuk perhitungan jumlah Riemann, yakni : **rightsum**, **rightbox**, **leftsum**, **leftbox**, dan **middlesum**, **middlebox** yang dapat menampilkan gambar dan jumlah Riemann secara otomatis. Meskipun demikian, Anda harus memahami bagaimana proses untuk melakukan perhitungan tersebut. Berikut adalah 4 langkah untuk menghitung jumlah Riemann:

1. Tentukan Δx , yakni lebar setiap subinterval dengan menggunakan variabel n .
2. Hitung nilai x_k , yakni partisi interval perhitungan jumlah Riemann.
3. Menghitung dan menyederhanakan jumlah Riemann (dengan menggunakan perhitungan jumlah kanan)
4. Menghitung limitnya untuk n mendekati tak berhingga.

Kita mulai dari contoh yang sama yang kita gunakan di atas. Mula-mula kita definisikan fungsi dan batas-batas interval integrasi.

```
> restart; with(plots): with(student);
> f:= x -> x^3-10*x^2+20*x+10; a:=-3; b:=5;
```

$$f := x \rightarrow x^3 - 10x^2 + 20x + 10$$

$$a := -3$$

$$b := 5$$

Selanjutnya kita ikuti keempat langkah di atas:

1. Tentukan Δx , yakni lebar setiap subinterval dengan menggunakan variabel n :

> Delta := (b-a)/n;

$$\Delta := 8 \frac{1}{n}$$

2. Hitung nilai x_k , yakni partisi interval perhitungan jumlah Riemann:

> x[k] := a + k*Delta;

$$x_k := -3 + \frac{8k}{n}$$

3. Menghitung dan menyederhanakan jumlah Riemann (dengan menggunakan perhitungan jumlah kanan):

> Sum(f(x[k])*Delta,k=1..n): % = simplify(value(%));

$$\sum_{k=1}^n \left(8 \frac{\left(-3 + \frac{8k}{n}\right)^3 - 10\left(-3 + \frac{8k}{n}\right)^2 - 50 + \frac{160k}{n}}{n} \right) = -\frac{8}{3} \frac{-228n + 224 + 49n^2}{n^2}$$


4. Menghitung limitnya untuk n mendekati tak berhingga:

> Limit(%%,n = infinity): % = evalf(value(%));

$$\lim_{n \rightarrow \infty} \sum_{k=1}^n \left(8 \frac{\left(-3 + \frac{8k}{n}\right)^3 - 10\left(-3 + \frac{8k}{n}\right)^2 - 50 + \frac{160k}{n}}{n} \right) = -130.6666667$$

Ternyata jumlah Riemannya negatif. Hal ini dapat diperjelas apabila kita gambar daerah antara kurva dan sumbu- x tersebut:

> rightbox(f(x),x=-3..5,20);



Ternyata daerah persegi panjangnya lebih luas yang berada di bawah sumbu- x , sehingga hasil perhitungannya negatif.

B. Beberapa Integral Menarik

Tujuan pembicaraan kita sekarang adalah untuk mempelajari bagaimana menghitung integral tentu dan integral tak tentu dengan menggunakan Maple serta untuk menelusuri berbagai sifat dan teorema integral.

> restart;

1. Integral Tentu dan Tak Tentu


Misalkan kita mempunyai fungsi sebagai berikut:

```
> f := x -> .01*x^4+3+sqrt(x)+6/x^2+x*sin(3*x);
```

$$f := x \rightarrow .01 x^4 + 3 + \sqrt{x} + \frac{6}{x^2} + x \sin(3 x)$$

Selanjutnya, kita gambar kurva fungsi tersebut dan batas-batas daerah yang akan kita hitung luasnya:

```
> plot({f(x),[[1,0],[1,f(1)]],[[6,0],[6,f(6)]]},x=0..7,y=5..20);
```



Dengan Maple kita dapat menghitung integral tentu maupun integral tak tentu. Perintah Maple untuk menghitung integral adalah **Int** (dengan huruf I besar) atau **int** (dengan i kecil). Perintah **Int** menghasilkan ekspresi integral tanpa menghitung nilainya, sedangkan perintah **int** menghitung anti-derivatif. Untuk menampilkan hasil perhitungan ekspresi integral dari perintah **Int** dapat digunakan perintah **value**. Perhatikan contoh perhitungan integral fungsi f di atas sebagai berikut:

```
> Int(f(x),x); value(%); int(f(x),x);
```

$$\int .01 x^4 + 3 + \sqrt{x} + \frac{6}{x^2} + x \sin(3 x) dx$$

$$.002000000000 x^5 + 3. x + .6666666667 x^{(3/2)} - \frac{6.}{x} + .1111111111 \sin(3. x) - .3333333333 x \cos(3. x)$$

$$.002000000000 x^5 + 3. x + .6666666667 x^{(3/2)} - \frac{6.}{x} + .1111111111 \sin(3. x) - .3333333333 x \cos(3. x)$$

Kedua perintah **Int** dan **int** juga dapat digunakan untuk menghitung integral tentu, yakni luas daerah di antara kurva dan sumbu- x . Dalam hal ini variabel integrasinya diberi nilai batas-batas integrasi. Perhatikan contoh di bawah ini, bandingkan dengan pemakaian kedua perintah sebelumnya.

```
> Int(f(x),x=1..6); value(%); int(f(x),x=1..6);
```

$$\int_1^6 .01 x^4 + 3 + \sqrt{x} + \frac{6}{x^2} + x \sin(3 x) dx$$

42.93153836

42.93153836

Seperti Anda lihat, perintah **Int** memerlukan perintah **value** untuk menampilkan hasil perhitungan,

sedangkan perintah `int` akan langsung menghasilkan nilai integralnya. Cobalah Anda lakukan perhitungan integral tak tentu dan integral tentu untuk fungsi-fungsi lain dengan batas-batas integrasi Anda tentukan sendiri!

2. Sifat Linieritas Integral

Suatu transformasi T yang memenuhi sifat $T(a f(x) + b g(x)) = a T(f(x)) + b T(g(x))$ disebut transformasi linier. Derivatif adalah salah satu contoh transformasi linier. Demikian pula, integral juga merupakan contoh transformasi linier yang lain. Marilah kita telusuri sifat-sifat tersebut dengan mendefinisikan dua buah fungsi f dan g .

> `f := x -> x^3+10*x+3;`

$$f := x \rightarrow x^3 + 10x + 3$$

> `g := x -> sin(10*x);`

$$g := x \rightarrow \sin(10x)$$

> `Int(a*f(x)+b*g(x),x=0..14) : % = value(%);`

$$\int_0^{14} a(x^3 + 10x + 3) + b \sin(10x) dx = 10626a - \frac{1}{10}b \cos(140) + \frac{1}{10}b$$

> `a*Int(f(x),x=0..14)+b*Int(g(x),x=0..14) : % = value(%);`

$$a \int_0^{14} x^3 + 10x + 3 dx + b \int_0^{14} \sin(10x) dx = 10626a + b \left(-\frac{1}{10} \cos(140) + \frac{1}{10} \right)$$

Anda melihat bahwa hasilnya keduanya sama. Hal itu memperlihatkan (namun bukan bukti formal) kebenaran sifat linieritas integral. Apabila nilai-nilai a dan b kita tentukan, maka hasilnya akan terlihat persis sama:

> `a:=3;b:=5;`

$$a := 3$$

$$b := 5$$

> `Int(a*f(x)+b*g(x),x=0..14) : % = value(%);`

$$\int_0^{14} 3x^3 + 30x + 9 + 5 \sin(10x) dx = \frac{63757}{2} - \frac{1}{2} \cos(140)$$

> `a*Int(f(x),x=0..14)+b*Int(g(x),x=0..14) : % = value(%);`

$$3 \int_0^{14} x^3 + 10x + 3 dx + 5 \int_0^{14} \sin(10x) dx = \frac{63757}{2} - \frac{1}{2} \cos(140)$$

Silakan Anda telusuri kebenaran sifat tersebut dengan menggunakan fungsi-fungsi dan konstanta-konstanta lain! Dapatkan Anda membuktikan kebenaran sifat tersebut secara formal (deduktif)? Silakan Anda lihat buku-buku kalkulus jika tidak yakin!

3. Batas Bawah dan Atas

Suatu konstanta M dikatakan batas atas sebuah fungsi apabila nilai fungsi tersebut pada suatu interval tidak pernah melebihi M , yakni $f(x) \leq M$. Secara serupa, suatu konstanta m dikatakan **batas bawah** sebuah fungsi apabila nilai fungsi tersebut pada suatu interval selalu melebihi m , yakni $m \leq f(x)$. Di

dalam kenyataan, terdapat beberapa fungsi yang sulit atau tidak mungkin diintegrasikan. Akan tetapi, dengan menggunakan batas atas atau batas bawah suatu fungsi, kita dapat memperoleh batas atas dan batas bawah integralnya.

Pada Maple terdapat paket `students` yang memuat perintah `maximize` dan `minimize` untuk menghitung nilai maksimum dan minimum suatu fungsi pada sebuah interval. Perhatikan contoh perhitungan di bawah ini.

```
> restart; with(student);
> g := x -> 1/3*x^3-7*x^2+35*x+30;
```

$$g := x \rightarrow \frac{1}{3}x^3 - 7x^2 + 35x + 30$$

```
> minimize(g(x),x=0..14); m:=evalf(%);
maximize(g(x),x=0..14); M:=evalf(%);
```

$$275 + \frac{1}{3}(7 + \sqrt{14})^3 - 7(7 + \sqrt{14})^2 + 35\sqrt{14}$$


$$m := 11.4111973$$

$$275 + \frac{1}{3}(7 - \sqrt{14})^3 - 7(7 - \sqrt{14})^2 - 35\sqrt{14}$$

$$M := 81.2554690$$

Berikut kita gambar kurva fungsi dan batas atas serta batas bawahnya.

```
> plot({g(x),m,M},x=0..14,y=0..90);
```



Garis-garis mendatar menunjukkan nilai-nilai minimum dan maksimum fungsi tersebut pada interval. Apabila m adalah batas bawah fungsi f pada interval $[a, b]$ dan M adalah batas atasnya, maka $m(b-a)$ merupakan luas daerah persegi panjang kecil yang berada di bawah kurva, dan $M(b-a)$ adalah luas persegi panjang besar yang memuat kurva. Secara numerik hal ini dapat ditunjukkan sebagai berikut.

```
> a:=0;b:=14;
```

$$a := 0$$

$$b := 14$$

```
> Int(m,x=a..b): % = value(%);
Int(g(x),x=a..b): % = evalf(value(%));
Int(M,x=a..b): % = value(%);
```

$$\int_0^{14} 11.4111973 \, dx = 159.7567622$$

$$\int_0^{14} \frac{1}{3}x^3 - 7x^2 + 35x + 30 dx = 648.6666667$$

$$\int_0^{14} 81.2554690 dx = 1137.576566$$

Anda dapat melihat batas bawah, batas atas, dan nilai integralnya. Dalam hal ini berlaku $m \leq g(x)$, $g(x) \leq M$, $m(b-a) \leq \int_a^b g(x) dx$, dan $\int_a^b g(x) dx \leq M(b-a)$. Jadi, apabila kita tidak dapat menghitung nilai $\int_a^b g(x) dx$ secara langsung, maka kita hanya dapat mengatakan bawah nilainya terletak antara $m(b-a)$ dan $M(b-a)$. Carilah sebuah fungsi yang tidak dapat diintegrasikan namun diketahui batas atas dan batas bawahnya pada suatu interval, kemudian perkirakan nilai integralnya!

4. Harga Mutlak dan Integral

Kita adakan menelusuri suatu proposisi yang menyatakan bahwa $\left| \int_a^b f(x) dx \right| \leq \int_a^b |f(x)| dx$ dan melihat kebenarannya. Mula-mula kita definisikan sebuah fungsi, kemudian kita hitung kedua nilai dan kita bandingkan mana yang lebih besar.

> `f := x -> x^3-10*x^2+20*x+10;`

$$f := x \rightarrow x^3 - 10x^2 + 20x + 10$$

Berikut adalah integral harga mutlak fungsi pada interval $[-3, 8]$:

> `Int(abs(f(x)),x=-3..8): % = evalf(value(%));`

$$\int_{-3}^8 |x^3 - 10x^2 + 20x + 10| dx = 282.6670755$$


Berikut adalah harga mutlak integral fungsi f pada interval yang sama:

> `abs(Int(f(x),x=-3..8)): % = evalf(value(%));`


$$-\int_{-3}^8 x^3 - 10x^2 + 20x + 10 dx = 132.9166667$$

Kita lihat hasilnya lebih kecil daripada hasil pertama. Apabila kita gambar daerah integrasinya, kita akan melihat mengapa kedua hasil berbeda, dengan mengingat kembali pengertian luas positif dan luas negatif.

> `middlebox(f(x),x=-3..8,500);`



> middlebox(abs(f(x)),x=-3..8,500);



Pada integral fungsi mungkin terdapat luas negatif, sehingga hasil integrasinya lebih besar, sedangkan pada integral harga mutlak semua luasnya positif, sehingga hasilnya lebih besar. Dapatkah Anda membuktikan proposisi tersebut secara formal? (Lihat buku-buku kalkulus!)

5. Teorema Dasar Kalkulus

Salah satu Teorema Dasar Kalkulus menyatakan bahwa jika $\int f(x) dx = F(x)$, maka $\frac{\partial}{\partial x} F(x) = f(x)$.

Kita akan menelusuri kebenaran teorema ini dengan mendefinisikan sebuah fungsi f , kemudian menentukan $F(x)$ dengan mengintegrasikan $f(x)$, dan terakhir menurunkan kembali $F(x)$ apakah sama dengan $f(x)$.

> f := x -> x^2 - 10*x + 30;

$$f := x \rightarrow x^2 - 10x + 30$$

> F := x -> int(f(t), t=0..x);

$$F := x \rightarrow \int_0^x f(t) dt$$

Dengan definisi F tersebut kita dapatkan:

> F(x);

$$\frac{1}{3}x^3 - 5x^2 + 30x$$

Apabila kita turunkan, akan diperoleh fungsi semula, yakni:

> diff(F(x), x);

$$x^2 - 10x + 30$$

Untuk menghitung fungsi yang lain, definisikan terlebih dahulu f , misalnya:

> f := x -> ln(x) + cos(x);

$$f := x \rightarrow \ln(x) + \cos(x)$$

> F(x);

$$x \ln(x) - x + \sin(x)$$


> diff(F(x), x);

$$\ln(x) + \cos(x)$$

Kita dapat menggambar kedua kurva $y = f(x)$ dan $y = F(x)$. Dalam hal ini kurva kedua dapat dipandang sebagai ukuran luas daerah di bawah kurva pertama dari 0 sampai x . Dapatkah Anda menunjukkan mana kurva f dan mana yang merupakan kurva F ? Perhatikan titik potongnya dengan sumbu- y !


> f := x -> x^2-10*x+30; plot({f(x),F(x)},x=0..10);

$$f := x \rightarrow x^2 - 10x + 30$$



> f:=x->ln(x)+cos(x); plot({f(x),F(x)},x=0..10);

$$f := x \rightarrow \ln(x) + \cos(x)$$



C. Benda Putar

Pada modul ini kita akan mengkaji benda-benda putar yang diperoleh dengan cara memutar kurva suatu fungsi di sekitar sumbu- x dan sumbu- y . Kita akan menggambar benda putar tersebut dan menghitung volumenya, baik secara simbolik maupun numerik.

> restart; with(plots):


1. Perputaran di Sekitar Sumbu- X

Marilah kita lihat bagaimana penampakan benda putar yang diperoleh dengan memutar kurva sebuah fungsi di sekitar sumbu- x . Mula-mula kita definsikan fungsinya dan kita gambar kurvanya.

> f := x -> 2-sin(x);

$$f := x \rightarrow 2 - \sin(x)$$


> plot(f(x),x = 0..2*Pi,y=0..3);



Kita akan memutar kurva tersebut di sekitar sumbu- x . Untuk menghasilkan pandangan tiga dimensi benda putar yang dihasilkan kita akan menggunakan sistem koordinat ruang (x,y,z) . Sumbu- z akan mengukur ketinggian permukaan bidang xy . Setiap titik $(x, f(x))$ pada kurva asli akan diputar mengitari sumbu- x menghasilkan sebuah lingkaran berjari-jari $f(x)$. Setiap titik pada lingkaran mempunyai titik koordinat berbentuk (a,b,c) dengan $a = x$, $b^2 + c^2 = f(x)^2$. Dari ide ini, kita dapat menggambar benda putar yang kita inginkan dengan menggunakan perintah **implicitplot3d** untuk fungsi implisit berbentuk $y^2 + z^2 = f(x)^2$ dengan menentukan batas-batas nilai-nilai x, y, z .


Perhatikan contoh berikut ini.

> implicitplot3d({y^2+z^2=f(x)^2},x=0..2*Pi,y=-3..3,z=-3..3);



Untuk menghasilkan gambar yang sesuai dengan keinginan kita dapat digunakan tambahan parameter:

> implicitplot3d({y^2+z^2=f(x)^2},x=0..2*Pi,y=-3..3,z=-3..3,
style=patchngrid,orientation=[-50,70],axes=boxed);



Pada layar Maple, Anda dapat mengklik gambar tersebut dan menekannya sambil memutar-mutar untuk melihat gambar dari berbagai posisi (arah), bahkan Anda dapat melihat bagian dalam benda tersebut! Berikut adalah contoh lain yang dapat Anda gunakan sebagai template untuk mengerjakan latihan-latihan Anda sendiri.


```
> f:= x -> 1-x^2+2*x;  a:=-1; b:= 1;
```

$$f := x \rightarrow 1 - x^2 + 2x$$


$$a := -1$$

$$b := 1$$

```
> plot(f(x),x=a..b);
```



```
> implicitplot3d(y^2+z^2=f(x)^2,x=a..b,y=-2..2,z=-2..2,
style=patchngrid,orientation=[-50,70],axes=boxed,numpoints=2000);
```



Cobalah Anda buat gambar benda-benda putar yang lain sesuai dengan keinginan Anda sendiri!


2. Perputaran di Sekitar Sumbu-Y

Dengan cara serupa kita juga dapat menggambar benda putar yang diperoleh dengan memutar kurva sebuah fungsi terhadap sumbu- y . Perbedaannya di sini dengan cara sebelumnya adalah persamaan lingkaran yang diperoleh dari perputaran setiap titik pada kurva adalah $x^2 + z^2 = g(y)^2$. Jadi, dalam hal ini, fungsi semula merupakan fungsi dalam y . Sebagai contoh, apabila $f(x) = x^2$, maka fungsi yang digunakan adalah $g(y) = \sqrt{y}$. Perhatikan contoh berikut ini.


```
> g := y -> sqrt(y);
```

$$g := \text{sqrt}$$

```
> implicitplot(x=g(y),x=-1..1,y=-1..1);
```

```
> implicitplot3d(x^2+z^2=g(y)^2,x=-1..1,y=0..1,z=-1..1,
style=patchngrid,orientation=[-90,-20],axes=normal,numpoints=2000);
```



Silakan Anda coba sendiri membuat gambar benda-benda putar yang lain!


3. Perputaran Daerah Antara Dua Kurva

Kita dapat memutar suatu daerah antara dua buah kurva berlainan. Hal ini agak rumit, namun prinsipnya sama dengan perputaran sebuah kurva. Perhatikan contoh di bawah ini (untuk perputaran terhadap sumbu- x).


```
> f:=x->x^2; g:=x->2-x; plot({f(x),g(x)},x=0..1.5);
```

$$f := x \rightarrow x^2$$

$$g := x \rightarrow 2 - x$$



```
> display(implicitplot3d(y^2+z^2=f(x)^2,x=0..1.5,y=-2..2,z=-2..2,
style=patchcontour,shading=xyz,numpoints=2000),
implicitplot3d(y^2+z^2=g(x)^2,x=0..1.5,y=-2..2,z=-2..2,
style=patchngrid,axes=boxed,shading=z,numpoints=2000,
orientation=[-145,55]));
```



Pada tampilan tersebut benda putar luar dan benda putar dalam digambar dengan warna yang berlainan. Dengan sudut pandang yang sesuai Anda akan memperoleh gambar benda putar yang dihasilkan.

4. Metode Cakram

Sekarang kita akan menghitung volume benda putar yang diperoleh dengan memutar kurva sebuah fungsi di sekitar sumbu- x dengan menggunakan metode cakram. Dalam metode ini, kita bayangkan potongan luas yang harus diputar menjadi lempeng-lempeng tipis tegak berukuran $y \Delta x$, kemudian memutarnya membentuk sebuah cakram yang dapat Anda bayangkan seperti sebuah roti mari atau koin tipis, kemudian menjumlahkan volume semua cakram untuk menghasilkan volume (integral)nya. Rumus

volume setiap cakram dalam hal ini adalah $dV = \pi y^2 dx$, sehingga $V = \int_a^b \pi y^2 dx$ dengan a dan b

atas batas-batas integrasi (interval perputarannya). Berikut adalah sebuah contoh. Kita akan memutar $y = 1 + x \sin(x)$ antara $x = \frac{\pi}{4}$ dan $x = \pi$ mengitari sumbu- x . Perhatikan langkah-langkah berikut ini.


> `f:=x->1+x*sin(x); a:=Pi/4; b:=Pi; # definisi fungsi dan batas-batas integrasi`

$$f := x \rightarrow 1 + x \sin(x)$$


$$a := \frac{1}{4} \pi$$

$$b := \pi$$

> `plot({f(x),0},x=a..b); # plot kurva fungsinya`



> `implicitplot3d(y^2+z^2=f(x)^2,x=a..b,y=-3..3,z=-3..3, style=patchngrid,axes=boxed,orientation=[-105,85], numpoints=2000); # gambar benda putar yang terjadi`



Selanjutnya, kita hitung volume benda putar dengan integralnya secara simbolik dan nilai eksaknya serta hasil numeriknya.

> `Int(Pi*f(x)^2,x=a..b): %=value(%);`

$$\int_{1/4\pi}^{\pi} \pi (1 + x \sin(x))^2 dx = \frac{5}{2} \pi^2 + \frac{21}{128} \pi^4 + \frac{1}{4} \pi^2 \sqrt{2} - \pi \sqrt{2} + \frac{1}{64} \pi^3 - \frac{1}{8} \pi$$

> `%%=evalf(value(%%));`

$$\int_{1/4\pi}^{\pi} \pi (1 + x \sin(x))^2 dx = 39.79351317$$

> `V:=%;`

$$V := \int_{1/4\pi}^{\pi} \pi (1 + x \sin(x))^2 dx = 39.79351317$$

Cobalah Anda lakukan perhitungan volume benda putar untuk fungsi-fungsi yang lain. Pertama, lakukan perhitungan volume benda putar terhadap sumbu- x , **kemudian** lakukan perhitungan volume benda putar terhadap sumbu- y .

5. Metode Cincin

Metode cakram di atas hanya dapat digunakan apabila lempeng-lempeng yang diputar kontinu tanpa ada jarak antara sumbu- x dan kurva, dan tidak dapat dipakai apabila terdapat jarak antara lempeng dan sumbu- x . Sebagai contoh adalah apabila kita ingin menghitung volume benda putar dari perputaran daerah antara dua buah kurva terhadap sumbu- x . Perhatikan contoh sebagai berikut.

> `f:=x->x^(1/8); # definisi kurva pertama`

$$f := x \rightarrow x^{(1/8)}$$

> `g:=x->x^8; # definisi kurva kedua`


$$g := x \rightarrow x^8$$

> `a:=0; b:=1; # batas-batas integrasi`

$$a := 0$$

$$b := 1$$

> `plot({f(x),g(x)},x=a..b); # plot kedua kurva`



Pada interval $[0, 1]$ kurva yang di atas adalah kurva $y = f(x)$, sehingga volume setiap cincin adalah $dV = \pi (f(x)^2 - g(x)^2) dx$, sehingga volume benda putar yang terjadi adalah $V = \pi \int_a^b (f(x)^2 - g(x)^2) dx$. Hal ini dapat dilakukan dengan mudah pada Maple.

> `Int(Pi*(f(x)^2-g(x)^2),x=a..b): %=value(%);`

$$\int_0^1 \pi (x^{(1/4)} - x^{16}) dx = \frac{63}{85} \pi$$

> `%%=evalf(value(%));`

$$\int_0^1 \pi (x^{(1/4)} - x^{16}) dx = 2.328474555$$

> `V:=%;`

$$V := \int_0^1 \pi (x^{(1/4)} - x^{16}) dx = 2.328474555$$

Catatan:

Rumus di atas tidak berlaku umum, Anda harus memperhatikan kedua kurva, mana yang di atas dan mana yang di bawah untuk menentukan mana pengurang dan mana yang dikurangi.

6. Metode Selimut Silinder

Terakhir kita akan menghitung volume benda putar yang diperoleh dengan memutar suatu kurva terhadap sumbu-ry dengan menggunakan metode selimut silinder. Dalam metode ini, benda butar dibagi menjadi silinder-silinder tegak dengan jari-jari x , ketinggian y , dan tebal dx , sehingga volume setiap silinder adalah $dV = 2 \pi x y dx$. Volume benda putar yang terjadi adalah $V = 2 \pi \int_a^b x f(x) dx$.

Perhatikan contoh berikut ini.


> `f:=x->sqrt(abs(x^2-2)); a:=0;b:=4;`

$$f := x \rightarrow \sqrt{|x^2 - 2|}$$

$$a := 0$$

$$b := 4$$

```
> plot({f(x),0},x=a..b);
```



Perhatikan, untuk contoh ini sangat sulit kita menghitung volume dengan menggunakan metode cincin sekalipun dengan bantuan Maple. Di sini Anda perlu mempertimbangkan sebuah cincin horisontal dengan rumus volume berbentuk seperti $(4^2 - s^2 + r^2) dx$ dengan r dan s keduanya ditentukan dari fungsi yang sama dengan rumus yang berbeda-beda untuk setiap nilai x . Untunglah hal itu lebih mudah dilakukan dengan menggunakan metode selimut silinder, karena rumus volume selimut silinder untuk setiap dx tidak berbeda dalam hal ini.

Berikut adalah perhitungan volume benda putarnya dengan menggunakan integral.

```
> Int(2*Pi*x*f(x),x = a..b): %=value(%);
```

$$\int_0^4 2 \pi x \sqrt{|x^2 - 2|} dx = \frac{28}{3} \sqrt{14} \pi + \frac{4}{3} \pi \sqrt{2}$$

```
> %%=evalf(value(%));
```

$$\int_0^4 2 \pi x \sqrt{|x^2 - 2|} dx = 115.6349686$$

```
> V:=%;
```

$$V := \int_0^4 2 \pi x \sqrt{|x^2 - 2|} dx = 115.6349686$$